COMPUTER ALGORITHM FOR THE ANALYSIS OF UNDERDAMPED AND OVERDAMPED WATER-LEVEL RESPONSES IN SLUG TESTS

By Nicasio Sepúlveda

U. S. GEOLOGICAL SURVEY

Water-Resources Investigations Report 91-4162

Prepared in cooperation with the

PUERTO RICO INDUSTRIAL DEVELOPMENT COMPANY

San Juan, Puerto Rico

1992

U.S. DEPARTMENT OF INTERIOR

MANUEL LUJAN, JR., Secretary

U.S. GEOLOGICAL SURVEY

Dallas L. Peck, Director

For additional information write to:

District Chief
U.S. Geological Survey
P.O. Box 364424
San Juan, Puerto Rico 00936-4424

Copies of this report can be purchased from:

U.S. Geological Survey
Books and Open-File Reports
Federal Center
Box 25425, Denver, Colorado 80225

CONTENTS

ILLUSTRATIONS

TABLES

CONVERSION FACTORS

Multiply	By	To obtain
	Length	
foot	0.3048	meter
	Hydraulic conductivity	
foot per day	0.3048	meter per day
	Transmissivity	
foot squared per day	0.09290	meter squared per day

COMPUTER ALGORITHM FOR THE ANALYSIS OF UNDERDAMPED AND OVERDAMPED WATER-LEVEL RESPONSES IN SLUG TESTS

By Nicasio Sepúlveda

ABSTRACT

A computer algorithm is presented for the analysis of two types of water-level responses commonly observed in slug tests on well-aquifer systems: underdamped and overdamped. General guidelines established in the literature for the analysis of slug-test data are implemented in the algorithm and are illustrated and discussed through the use of field data. The theory for the underdamped response, treated by the van der Kamp approximate solution, and the overdamped response, treated by the Bouwer and Rice approximate solution, is briefly reviewed. The algorithm provides a technique to eliminate the inclusion, in the slug-test analysis, of sinusoidal peaks in a water-level response that have an angular frequency and damping constant that differ substantially from those of other peaks in an underdamped response. These anomalous sinusoidal peaks generally occur for early-time data, when the head in the well does not change instantaneously and the assumptions on which van der Kamp solution is based are not met. The input and output files for the computer algorithm are described in this report.

INTRODUCTION

The water-level oscillations noted during slug tests of wells completed in fractured, consolidated rocks is analogous to the oscillations of a mass on a spring in a viscous medium: the water in the well corresponds to the mass and the aquifer corresponds to the spring. Depending on the mass of water in the well and the hydraulic conductivity of the aquifer, the response of the water level in the well may behave as an overdamped, critically damped, or underdamped oscillator. When the well-aquifer system behaves as an overdamped oscillator, the water level slowly returns to the equilibrium level. The critical damping response, not addressed in this report, is the transition between the two responses. Kabala and others (1985) as well as Kipp (1985) have addressed the critical damping response. The underdamped response is typical of water-bearing zones of relatively high hydraulic conductivity.

The subjectivity inherent in the selection of sinusoidal peaks used for the analysis of an underdamped water-level response generally causes errors in the calculation of the hydraulic conductivity of an aquifer. The computer program presented here establishes guidelines to correctly identify and exclude from the slug-test analysis such sinusoidal peaks.

Purpose and Scope

This report presents a computer program that can be used to analyze both underdamped and overdamped water-level responses to a slug test. The methodologies outlined in this report will assist hydrologists in accurately analyzing slug-test data. Although the calculation of hydraulic conductivity from slug-test data of an overdamped water-level response is straight forward and computer programs have been developed to analyze this type of response, the computer program described in this report includes the algorithm for an overdamped water-level response to provide a complete treatment of both responses. However, it is the analysis of the underdamped water-level response that is improved upon in this computer program.

Theoretical Review of Slug-Test Analysis

The principle of a slug test lies in the instantaneous removal or injection of a mass of water in a well that partially or fully penetrates an aquifer. The time-dependent water-level response inside the well casing following this removal or injection of a mass, or slug, of water constitutes the slug-test data. If the water-level changes are of the form of a continuous decay (in the case of an injection) or a continuous recovery (in the case of a removal) towards the initial water level or equilibrium condition, then the water-level response is called an overdamped response. However, if the water-level changes oscillate around the initial equilibrium level, then the water-level response is characterized as underdamped.

van der Kamp (1976) developed an algorithm, for the underdamped water-level response, to estimate the transmissivity of an aquifer from slug-test data. The development of the algorithm for the overdamped water-level response, to estimate the hydraulic conductivity of an aquifer near a well from slug-test data, was done by Bouwer and Rice (1976). These two algorithms make it possible to obtain estimates of hydraulic conductivity from slug-test data for most aquifers.

The parameters used in van der Kamp's algorithm (1976), to estimate the transmissivity T (in feet squared per day) for the underdamped response, are the casing radius, r_c, the well screen radius (including well-bore), r_s, and the well screen length, h. van der Kamp analysis assumes that the well screen is open to the full aquifer thickness. If this condition is not met, the test results may reflect the hydraulic conductivity of the screened section of the aquifer. The parameters used in the Bouwer and Rice (1976) algorithm, to estimate the hydraulic conductivity K (in feet per day) for the overdamped response, are H and D, which are the distances from the water table to the bottom of the screen and to the bottom of the aquifer, respectively. All of these parameters are illustrated in figure 1. The value of specific storage, S_s, must be known in order to apply van der Kamp's (1976) algorithm. However, the Bouwer and Rice algorithm (1976) does not consider the effect of specific storage.



Figure 1.-- Geometry and notation used for the analysis of underdamped and overdamped water-level responses in slug tests.

Underdamped Response

The underdamped response is characterized by water levels oscillating around the equilibrium level. An example of a characteristic underdamped response is shown in figure 2. This response was analyzed using the approximation theory developed by van der Kamp (1976). This theory is based on the assumption that the changes in water level, after the equivalent of a given mass of water is removed or added to the well-aquifer system, are dictated by the equation

$$W(t) = A_0 \, e^{-\gamma t} \cos(\omega t), \tag{1}$$

where

$W(t)$ is the water level as a function of time, in feet;

A_0 is the initial displacement of the water level, in feet, assuming zero is the equilibrium position;

ω is the angular frequency, in seconds $^{-1}$(s^{-1}); and,

γ is the damping factor, in s^{-1}.

van der Kamp's (1976) algorithm is based on the assumption that ω and γ remain constant throughout a given response. This implies that the sinusoidal peaks characteristic of the underdamped response should all have approximately the same values of angular frequency and damping constants. If the sinusoidal peaks have substantially different values of angular frequency, then the slug-test data is considered unusable.

The nonlinear equation to be solved for T in the underdamped response is:

$$T = -2.3026 \frac{r_c^2\ (\omega^2 + \gamma^2)}{8\gamma} \log_{10}\left(0.79 r_s^2 (S/T)\ \sqrt{(\omega^2+\gamma^2)}\right), \tag{2}$$

where S is the storage coefficient. The constant 2.3026 is the factor used to convert from natural logarithm to base 10 logarithm. S is the product of specific storage S_s and well screen length h (figure 1). The hydraulic conductivity, K, is obtained from $K = T/h$. After calculating the implicit derivatives $dT/d\gamma$ and $dT/d\omega$ from equation 2, it can be shown that $dT/d\gamma > 0$ if $\omega < C\gamma$, and $dT/d\gamma < 0$ if $\omega > C\gamma$, where C is a positive constant. This implies that if the damping constant γ is larger than the angular frequency ω, T will increase as γ increases. Also, as the angular



Figure 2.-- Underdamped response for data shown in table 1.

frequency ω increases (shorter time intervals between sinusoidal peaks), transmissivity T increases. The conditions for an underdamped response are satisfied if the square root of the argument of the base 10 logarithm in equation 2, divided by r_s, is much smaller than 0.1 (van der Kamp, 1976, p. 73). This condition can be satisfied if α, defined by $\alpha = 0.89(S/T)^{1/2}(\omega^2+\gamma^2)^{1/4} r_s$, is much smaller than 0.1. The constant C, introduced earlier, is given by $\sqrt{(1+2\ln\alpha)/(2\ln\alpha)}$.

Overdamped Response

The overdamped water-level response is characterized by exponential decay or recovery to equilibrium level. An example of a characteristic overdamped response is shown in figure 3. This response was analyzed using the algorithm developed by Bouwer and Rice (1976). This algorithm, based on the Thiem equation of steady-state flow to a well, is based on the assumption that if $W(t)$ is the water level as a function of time t; the plot of $\log_{10} W(t)$ versus t should yield a straight line. The hydraulic conductivity value K is obtained from the equation

$$K = (2.3026)^2 \frac{r_c^2 \log_{10}(R_e/r_s)}{2h} \frac{\log_{10}(W_0/W_t)}{(t-t_0)}, \tag{3}$$

where

W_0 is $W(t)$ at $t = t_0$;

W_t is $W(t)$; and

R_e is the radial distance at which the change in head, w in figure 1, caused by the slug-test perturbation becomes negligible.



Figure 3.-- Overdamped response for data shown in table 2.

SOLUTION ALGORITHM - COMPUTER PROGRAM

The algorithm implemented in the computer code for the calculation of the hydraulic conductivity of an underdamped water-level response is based on the computation of the angular frequency, ω; the maximum of the sinusoidal peak, W_0; and the time t_0 when this sinusoidal peak occurs. These three unknowns: W_0, ω, and t_0, are obtained from the solution of the following set of three nonlinear equations:

$$W_i = W_0 \cos(\omega(t_i - t_0)),\ i = 1,2,3, \tag{4}$$

where the pairs (t_1, W_1), (t_2, W_2), (t_3, W_3) are the three slug-test data points characterizing a single peak water level (see figure 2).

General Application

The data pairs of equation 4 for the underdamped condition are chosen so that W_1 and W_3 are smaller in absolute magnitude than W_2. These pairs are also contiguous with $t_1 < t_2 < t_3$. Because the term $e^{-\gamma t}$ in equation 1 is an asymptote that modulates the damping of the sinusoid, it has the effect of damping the relative amplitude of consecutive sinusoidal peaks. It can safely be assumed that the solution to equation 4 is very close to the maximum of the sinusoidal peak, the angular frequency associated with the peak (not relative to the next sinusoidal peak), and the time at which the maximum peak occurred. Equation 4, solved numerically using the Levenberg-Marquardt algorithm (IMSL, 1987), with an analytic calculation of the Jacobian, is solved for every sinusoidal peak with a distinguishable shape. For the purpose of this report, a sinusoidal peak has a distinguishable shape if three data pairs; (t_1, W_1), (t_2, W_2), and (t_3, W_3); meet the requirements described in equation 4 and can be recognized within the slug-test data (see figure 2). The convergence of the iterative process required to solve equation 4 depends on the initial values, which are generated from the input data. Sinusoidal peaks above and below the equilibrium water level are considered in this computer code. The location of the three data pairs that form a sinusoidal peak can be recognized directly from the time-drawdown list of values.

The theory of slug tests is based on the assumption that the injection or the removal of a slug of water occurs instantaneously, which, in practice, is impossible. The algorithm presented here will help determine whether or not the first sinusoidal peak has similar hydraulic characteristics to the remaining sinusoidal peaks in the slug-test data set. Sometimes, the three data pairs that compose the first sinusoidal peak cannot be fit into a sinusoid. This can be addressed by discarding the data from the first peak and considering the remaining ones, which constitutes a significant correction to the computed hydraulic-conductivity value.

The time intervals at which water levels are measured are fixed for most data loggers. Even when the time intervals can be varied, it is very difficult to synchronize the timing of measurements and water-level response adequately to identify the maximum amplitude of a sinusoidal peak when it occurred. Solving equation 4 is equivalent to finding the optimal time shift in the argument of cos (ωt) in equation 1, that will permit the identification of the real maximum amplitude of a sinusoidal peak and the time at which it occurred.

The ω values obtained from the solution of equation 4 for all sinusoidal peaks are used to obtain an average value for the angular frequency. If ($t_{01,}$ W_{01}) and ($t_{02,}$ W_{02}) are the time and maximum water levels of two sinusoidal peaks of the same sign with respect to the equilibrium water level, and $t_{01} < t_{02}$, then $\log_{10}(W_{01}/W_{02})/(t_{02} - t_{01})$ is one of the values used in the calculation of an average value for the damping factor, γ. Other values are obtained in the same way from the remaining peaks above or below the equilibrium water level. Newton's method (Isaacson and Keller, 1966, p. 98) was used to solve equation 2 for T.

The calculation of the hydraulic conductivity of an overdamped response is based on a least-squares linear-regression fit of a subset of the slug-test data. The linear regression is based on data over one natural logarithmic cycle. Early-time data is used over the logarithmic cycle with the highest correlation coefficient. The term $\log_{10}(R_e/r_s)$ in equation 3 was obtained by discretizing the curves in figure 3 of Bouwer and Rice paper (1976). An Akima cubic spline interpolant (Akima, 1970) was used on the discrete data to obtain an output value for every input value of h/r_s. The slope obtained from the linear regression, the value of $\log_{10}(R_e/r_s)$ obtained from the cubic spline interpolation, and the values of r_c and h determine the value of K according to equation 3.

The computer program listed in appendix 1, written in Fortran 77, can be run on a personal computer or on a mainframe. The code requires a linkage to IMSL (1987) libraries while it is being compiled. The first input data file of the computer program, SLUG.IN1, contains the necessary slug-test data to obtain K from equation 3 or T from equation 2. A second input file, SLUG.IN2, listed in appendix 2, contains the values used by the cubic spline interpolant for the overdamped response.

Case Studies - Computation of Hydraulic Conductivity

Computer program input and output data files for the two examples of underdamped and overdamped responses shown in figures 2 and 3 are shown in figures 4 and 5, respectively. Field data for these tests are listed in tables 1 and 2. The first card image of the input file consists of eight values in free format. The first number of this card is a "1" if the data represents an underdamped response, or a "2'' if the data represents an overdamped response. The following numbers of the first card represent r_s, r_c, H, D, and h, respectively. The seventh value of the first card image is the reference water level of the pressure transducer at the beginning of the slug test. This is the water level recorded by the data logger at time $t = 0$ during the slug test. This water level is considered to represent equilibrium conditions. The last value of the first card is the number of sinusoidal peaks to be read if the response is underdamped or the number of data points to be read if the response is overdamped. The card images of the input data file that follow include those that contain the slug-test data points, three data pairs per line if sinusoidal peaks are being read (for an underdamped response, table 1) or the entire set of data pairs, one pair per line, for the overdamped response (table 2). The last card image of the input data file, for the underdamped response, contains the specific storage value S_s first, and then the threshold value for the convergence of equation 4. A threshold value of 0.00005 is suggested to require convergence of the solution to equation 4.

Input data file: SLUG.IN1

```
1   0.1670    0.06250    127.540    241.540    27.0    -0.04    4
0.0833    -1.05        0.1000    -1.29       0.1166    -1.22
0.1833     0.68        0.2000     0.92       0.2166     0.89
0.2833    -0.59        0.3000    -0.80       0.3166    -0.80
0.3333    -0.62        0.4167     0.56       0.5000    -0.49
0.0000030     0.000050
```

Output data file: SLUG.OUT

```
APPLYING VAN DER KAMP METHOD ...
DAMPING VALUES ARE:  0.0386   0.0362
ANGULAR FREQUENCY VALUES ARE: 0.5034  0.5367  0.5317  0.5206
K  FROM VAN DER KAMP METHOD IS     122. FT/DAY
```

Screen display:

```
APPLYING VAN DER KAMP METHOD ...
S MAX = -1.2500   S FREQ =  0.5236   S TIME =  6.0000
F MAX = -1.2621   F FREQ =  0.5034   F TIME =  6.2755
S MAX =  0.9600   S FREQ =  0.5236   S TIME = 12.0000
F MAX =  0.9814   F FREQ =  0.5367   F TIME = 12.3901
S MAX = -0.7600   S FREQ =  0.4487   S TIME = 18.0000
F MAX = -0.7874   F FREQ =  0.5317   F TIME = 18.4980
S MAX =  0.6000   S FREQ =  0.5317   S TIME = 25.0020
F MAX =  0.6131   F FREQ =  0.5206   F TIME = 25.3992
DAMPING VALUES ARE:  0.0386   0.0362
ANGULAR FREQUENCY VALUES ARE:  0.5034   0.5367   0.5317   0.5206
K  FROM VAN DER KAMP METHOD IS     122. FT/DAY
```

Figure 4.-- Computer input and output files for the underdamped response shown in table 1.

Input data file: SLUG.IN1

```
2   0.1670    0.06250   286.110   310.110   12.0   0.00   17
0.0666       1.78
0.0833       1.51
0.1000       1.30
0.1166       1.11
0.1333       0.95
0.1500       0.79
0.1666       0.68
0.1833       0.54
0.2000       0.45
0.2166       0.38
0.2333       0.30
0.2500       0.26
0.2666       0.20
0.2833       0.17
0.3000       0.15
0.3166       0.11
0.3333       0.09
0.4167       0.03
0.5000       0.02
0.5833       0.01
0.6667       0.01
0.7500       0.01
0.8333       0.01
0.9167       0.01
1.0000       0.01
```

Output data file: SLUG.OUT

```
APPLYING BOUWER AND RICE METHOD ...
NUMBER OF POINTS USED IN LINEAR REGRESSION:  7
K  FROM BOUWER AND RICE METHOD IS      9.4 FT/DAY
```

Figure 5.-- Computer input and output files for the overdamped response shown in table 2.

The output file for the underdamped response consists of a line of text indicating the damping values, a second line of text indicating the angular frequency values, and another line of text indicating the hydraulic-conductivity value (figure 4). The screen display is different from the output data file and includes two lines of text for each sinusoidal peak that list the starting (S) and final (F) values (used in the solution of equation 4 for the maximum water level, the angular frequency, and the time at which the maximum water level occurred for each sinusoidal peak (four peaks in this example). The initial values used in the convergence process are computed from the input data. The final values listed are the maximum water level of the sinusoidal peak, in feet, the angular frequency, in s^{-1}, and the time at which the maximum water level occurred, in seconds. In case α is not much smaller than 0.1, a warning statement is printed as the last line of the output file.

For the underdamped response, the amplitude of the sinusoidal peaks decreases as time increases, but the damping as well as the angular frequency values should remain approximately constant. This is because the time separation between water-level peaks as well as the decay rate of the water-level amplitude should be approximately constant. Sinusoidal peaks corresponding to negative angular frequency values should be eliminated from the input file and the program must be run again with the new input file.

The output file for the overdamped response consists of a line of text indicating the number of points used in the linear regression and a second line of text indicating the hydraulic conductivity value (figure 5). The screen display for this response is the same as the output data file.

If the sinusoidal peak of the data listed in table 1, formed by the data pairs (0.0099, -0.04), (0.013, 4.53), and (0.0166, 3.30) had been included in the input data file as the first peak, then the angular frequency of that peak would have been 5.9403 s^{-1}, more than 10 times that of any one of the four angular frequencies listed in figure 4: 0.5034, 0.5367, 0.5317, and 0.5206 s^{-1}. If the peak formed by the pairs (0.0233, 0.53), (0.0266, 1.80), and (0.0300, 1.08) had been included in the input data file as the first peak, then the angular frequency would have been 5.4485 s^{-1}. If this angular frequency had been used in the average calculation for ω, an unreasonably high hydraulic conductivity value of about 917 feet per day would have been obtained. The solution shown in figure 4 was obtained without using these two peaks. Only the four sinusoidal peaks listed in figure 4, indicated by cards 2 through 5, and shown in figure 2, were used as input data for the final solution. The two peaks discarded (figure 2) were excluded because their characteristics were not consistent with the angular frequency values of the remaining four peaks. Their presence in the data is due to the inability to instantaneously remove or inject a slug of water. If first-peak data cause a lack of convergence when equation 4 is solved, they should be discarded and the next peak in the time series should be used as the first peak of the input data. Another cause of a lack of convergence in the solution of equation 4 is a poor definition of a given sinusoidal peak. This

Table 1.-- *Example of slug-test data for an underdamped water-level response*

Time, in minutes	Water level, in feet
0.0033	-0.04
0.0066	-0.04
0.0099	-0.04
0.0133	4.53
0.0166	3.30
0.0200	2.15
0.0233	0.53
0.0266	1.80
0.0300	1.08
0.0333	1.03
0.0500	0.32
0.0666	-0.50
0.0833	-1.05
0.1000	-1.29
0.1166	-1.22
0.1333	-0.86
0.1500	-0.30
0.1666	0.25
0.1833	0.68
0.2000	0.92
0.2166	0.89
0.2333	0.61
0.2500	0.22
0.2666	-0.22
0.2833	-0.59
0.3000	-0.80
0.3166	-0.80
0.3333	-0.62
0.4167	0.56
0.5000	-0.49
0.5833	0.14
0.6667	0.03
0.7500	-0.26
0.8333	0.20
0.9167	-0.25
1.0000	0.06
1.0833	-0.05
1.1667	-0.10
1.2500	0.05
1.3333	-0.13
1.4166	0.01
1.5000	-0.06
1.5833	-0.06
1.6667	-0.01
1.7500	-0.08
1.8333	-0.01
1.9167	-0.06
2.0000	-0.04
2.5000	-0.03
3.0000	-0.04
3.5000	-0.04
4.0000	-0.04

Table 2.-- *Example of slug-test data for an overdamped water-level response*

Time, in minutes	Water level, in feet
0.0033	0.00
0.0066	0.00
0.0099	0.00
0.0133	0.00
0.0166	0.00
0.0200	0.00
0.0233	0.00
0.0266	0.00
0.0300	0.00
0.0333	0.00
0.0500	2.69
0.0666	1.78
0.0833	1.51
0.1000	1.30
0.1166	1.11
0.1333	0.95
0.1500	0.79
0.1666	0.68
0.1833	0.54
0.2000	0.45
0.2166	0.38
0.2333	0.30
0.2500	0.26
0.2666	0.20
0.2833	0.17
0.3000	0.15
0.3166	0.11
0.3333	0.09
0.4167	0.03
0.5000	0.02
0.5833	0.01
0.6667	0.01
0.7500	0.01
0.8333	0.01
0.9167	0.01
1.0000	0.01
1.0833	0.02
1.1667	0.01

situation commonly occurs when apparent maximum water-level values cannot be ascertained from the data set because of poor time-of-measurement resolution.

SUMMARY

The computer algorithm presented in this report was developed, tested, and implemented for underdamped and overdamped water-level response conditions commonly encountered in slug tests of wells. The algorithm implemented for the underdamped response, based on the calculation of average angular frequency and damping constant values from the sinusoidal peaks of the slug test, allows for a more accurate calculation of the hydraulic conductivity by eliminating subjectivity in the selection of data needed for the analysis. The user has to evaluate the differences in angular frequency values to determine the sinusoidal peaks that should be excluded from the analysis. The algorithm implemented for the overdamped response, based on the calculation of the slope of the linear regression fit of water-level data over one natural logarithm cycle, makes use of a cubic spline interpolation to compute the hydraulic conductivity.

This computer code is expected to provide hydrologists with an improved method for analyzing underdamped water-level responses in slug tests. The analysis for the overdamped response has been provided in the computer code presented here as a matter of convenience. However, the analysis of overdamped water-level responses does not represent an improvement to existing methods of analysis.

REFERENCES

Akima, Hiroshi, 1970, A new method of interpolation and smooth curve fitting based on local procedures: Journal of the Association for Computing Machinery, v. 17, p. 589-602.

Bouwer, Herman and Rice, R.C., 1976, A slug test for determining hydraulic conductivity of unconfined aquifers with completely or partially penetrating wells: Water Resources Research, v. 12, no. 3, p. 423-428.

IMSL, 1987, User's manual: Math/Library Fortran Subroutines for Mathematical Applications, Houston, Texas, 1152 p.

Isaacson, Eugene and Keller, H.B., 1966, Analysis of numerical methods: New York, New York, John Wiley and Sons, Inc., 541 p.

Kabala, Z.J., Pinder, G.F., and Milly, P.C.D., 1985, Analysis of the well-aquifer response to a slug test: Water Resources Research, v. 21, no. 9, p. 1433-1436.

Kipp, Kenneth, Jr., 1985, Type curve analysis of inertial effects in the response of a well to a slug test: Water Resources Research, v. 21, no. 9, p. 1397-1408.

van der Kamp, Garth, 1976, Determining aquifer transmissivity by means of well response tests: the underdamped case: Water Resources Research, v. 12, no. 1, p. 71-77.

Appendix 1. Fortran 77 program listing

```
C
C     Written by Nicasio Sepulveda, hydrologist, USGS-WRD, San Juan, P.R.
C     Tel. (809) 749-4346 x 273  or  FTS 498-4346 x 273
C     This Fortran 77 computer program calculates the hydraulic conductivity
C     of an aquifer near a well based on slug-test data for the underdamped
C     and overdamped water-level responses.  The program uses Bouwer and Rice
C     (1976) algorithm to estimate the hydraulic conductivity for the
C     overdamped response using cubic splines to obtain values of curves
C     A,B, and C of fig. 3 in their paper.  The underdamped response is
C     analyzed following Van der Kamp's (1976) algorithm and a scheme
C     introduced to distinguish sinusoidal peaks with angular frequency
C     and damping constant values that deviate from those characteristic
C     values of the test, allowing for re-runs without including these
C     sinusoidal peaks.  Program makes IMSL calls for which this
C     subroutine library should be linked to.
C
C     NDAT   Number of cubic spline points used in Bouwer & Rice paper
C     NINTV  Number of cubic spline intervals used in Bouwer & Rice pape
C     B0,B1  Intercept and slope for early-time data of overdamped response
C     CSCOEF Matrix with cubic spline interpolation coefficients
C     XD     Vector with abscissa points of fig. 3 of Bouwer & Rice paper
C     FDA    Vector with ordinate points of curve A of fig. 3 of Bouwer & Ric
C     FDB    Vector with ordinate points of curve B of fig. 3 of Bouwer & Rice
C     FDC    Vector with ordinate points of curve C of fig. 3 of Bouwer & Rice
C     REFH   Reference water-level equilibrium datum
C     TI     Vector with time values at water-level measurements (underdamped)
C     HI     Vector with water-level values (underdamped)
C     HEI    Vector with computed peak-amplitude values (underdamped)
C     OME    Vector with angular frequency values of peaks (underdamped)
C     TIM    Vector with times at which each peak occurred (underdamped)
C     GAM    Vector with damping constant values (underdamped)
C     XVAL   Vector with abscissa values for overdamped response data
C     YVAL   Vector with ordinate values for overdamped response data
C     XIN    Vector with initial solution for equation (4)
C     X      Vector with final soultion for equation (4)
C     VAL    Common with amplitude HVAL and time TVAL used by IMSL
C
      PARAMETER (NDAT=55, NINTV=NDAT-1, NUM=200, N=3)
      REAL BREAK(NDAT),CSCOEF(4,NDAT)
      REAL XD(NDAT),FDA(NDAT),FDB(NDAT),FDC(NDAT)
      REAL B0,B1,STAT(12),REFH
      REAL TI(18),HI(18),HEI(6),OME(6),TIM(6),GAM(6)
      REAL XVAL(NUM),YVAL(NUM),X(N),XIN(N),HVAL(N),TVAL(N)
      INTEGER ID,INDEX
      EXTERNAL CSAKM,CSVAL,FCN,LSJAC,NEQNJ,RLINE
      COMMON /VAL/ HVAL,TVAL
C
      OPEN(31,FILE='SLUG.IN1',STATUS='OLD',ACCESS='SEQUENTIAL')
      OPEN(32,FILE='SLUG.IN2',STATUS='OLD',ACCESS='SEQUENTIAL')
      OPEN(33,FILE='SLUG.OUT',STATUS='UNKNOWN',ACCESS='SEQUENTIAL')
C
      G = 9.806650/0.30480
      PI = 4.0*ATAN(1.0)
C
C     Read eight values in first card image of input file
C
      READ(31,*) ID,RS,RC,H,DBR,RLBR,REFH,INDEX
C
C     Beginning of algorithm for the underdamped response
C
      IF(ID .EQ. 1) THEN
      WRITE(1,*) 'APPLYING VAN DER KAMP METHOD ... '
      WRITE(33,*) 'APPLYING VAN DER KAMP METHOD ... '
```

```
C
C     Read three (time, water-level) pairs for each sinusoidal peak
C     Time unit is converted from minutes to seconds
C     Water-level equilibrium is shifted to zero
C
      DO 2  I=1,INDEX
      READ(31,*) TI(3*I-2),HI(3*I-2),TI(3*I-1),HI(3*I-1),TI(3*I),HI(3*I)
      TI(3*I-2) = 60.0*TI(3*I-2)
      TI(3*I-1) = 60.0*TI(3*I-1)
      TI(3*I) = 60.0*TI(3*I)
      HI(3*I-2) = HI(3*I-2) - REFH
      HI(3*I-1) = HI(3*I-1) - REFH
      HI(3*I) = HI(3*I) - REFH
2     CONTINUE
C
C     Read a specific storage value
C
      READ(31,*) SPESTO,RELERR
      THICK = RLBR
      STO = SPESTO*RLBR
C
C     Calculate initial values for maximum water levels, angular frequency,
C     and time at which the maximum occurred, strictly from input data
C
      INDM1 = INDEX - 1
      DO 6  I=1,INDM1
      XIN(1) = HI(3*(I-1)+2)
      XIN(2) = PI/(TI(3*I+2) - TI(3*(I-1)+2))
      XIN(3) = TI(3*(I-1)+2)
      DO 4  K=1,N
      HVAL(K) = HI(3*(I-1)+K)
      TVAL(K) = TI(3*(I-1)+K)
4     CONTINUE
C
C     Solve nonlinear system of equations using subroutine NEQNJ of IMSL
C     Save the solution of the 3 by 3 system
C
      CALL NEQNJ(FCN,LSJAC,RELERR,N,200,XIN,X,FNORM)
      HEI(I) = X(1)
      OME(I) = X(2)
      TIM(I) = X(3)
      WRITE(1,98) XIN(1),XIN(2),XIN(3)
      WRITE(1,100) X(1),X(2),X(3)
6     CONTINUE
C
C     The last sinusoidal peak is analyzed separately
C
      XIN(1) = HI(3*INDM1 + 2)
      XIN(2) = X(2)
      XIN(3) = TI(3*INDM1 + 2)
      DO 8  K=1,N
      HVAL(K) = HI(3*INDM1 + K)
      TVAL(K) = TI(3*INDM1 + K)
8     CONTINUE
C
      CALL NEQNJ(FCN,LSJAC,RELERR,N,200,XIN,X,FNORM)
      HEI(INDEX) = X(1)
      OME(INDEX) = X(2)
      TIM(INDEX) = X(3)
      WRITE(1,98) XIN(1),XIN(2),XIN(3)
      WRITE(1,100) X(1),X(2),X(3)
C
C     Calculate the damping constants from the solution of the
C     nonlinear system of equations
C
```

```
      IF(INDEX .EQ. 2) THEN
      GAM(1) = ALOG(-HEI(1)/HEI(2))/(TIM(2) - TIM(1))
      GAMMA = GAM(1)
      WRITE(1,102) GAM(1)
      WRITE(33,102) GAM(1)
      ELSE IF(INDEX .EQ. 3) THEN
      GAM(1) = ALOG(HEI(1)/HEI(3))/(TIM(3) - TIM(1))
      GAMMA = GAM(1)
      WRITE(1,102) GAM(1)
      WRITE(33,102) GAM(1)
      ELSE IF(INDEX .EQ. 4) THEN
      GAM(1) = ALOG(HEI(1)/HEI(3))/(TIM(3) - TIM(1)
      GAM(2) = ALOG(HEI(2)/HEI(4))/(TIM(4) - TIM(2))
      GAMMA = 0.50*(GAM(1) + GAM(2))
      WRITE(1,102) GAM(1),GAM(2)
      WRITE(33,102) GAM(1),GAM(2)
      ELSE IF(INDEX .EQ. 5) THEN
      GAM(1) = ALOG(HEI(1)/HEI(3))/(TIM(3) - TIM(1))
      GAM(2) = ALOG(HEI(3)/HEI(5))/(TIM(5) - TIM(3))
      GAM(3) = ALOG(HEI(2)/HEI(4))/(TIM(4) - TIM(2))
      GAMMA = (GAM(1) + GAM(2) + GAM(3))/3.0
      WRITE(1,102) GAM(1),GAM(2),GAM(3)
      WRITE(33,102) GAM(1),GAM(2),GAM(3)
      ELSE IF(INDEX .EQ. 6) THEN
      GAM(1) = ALOG(HEI(1)/HEI(3))/(TIM(3) - TIM(1))
      GAM(2) = ALOG(HEI(3)/HEI(5))/(TIM(5) - TIM(3))
      GAM(3) = ALOG(HEI(2)/HEI(4))/(TIM(4) - TIM(2))
      GAM(4) = ALOG(HEI(4)/HEI(6))/(TIM(6) - TIM(4))
      GAMMA = 0.250*(GAM(1) + GAM(2) + GAM(3) + GAM(4))
      WRITE(1,102) GAM(1),GAM(2),GAM(3),GAM(4)
      WRITE(33,102) GAM(1),GAM(2),GAM(3),GAM(4)
      END IF
C
C     Obtain an average value of angular frequency from the
C     values obtained for the individual sinusoidal peaks
C
      OMEGA = 0.0
      DO 10  I=1,INDEX
      OMEGA = OMEGA + OME(I)
10    CONTINUE
      OMEGA = OMEGA/FLOAT(INDEX)
C
C     Solve nonlinear equation 2 in text for transmissivity
C
      RLVDK = G/(GAMMA**2 + OMEGA**2)
      WRITE(1,104) (OME(K), K=1,INDEX)
      WRITE(33,104) (OME(K), K=1,INDEX)
      DVDK = GAMMA/SQRT(G/RLVDK)
      A = RC*RC*SQRT(G/RLVDK)/(8.0*DVDK)
      B = -A*ALOG(0.790*RS*RS*STO*SQRT(G/RLVDK))
C
C     Use Newton's method to solve nonlinear transmissivity equation
C
      MAXIT = 20
      ITER = 0
      TOLD = B12     ITER = ITER + 1
      IF(ITER .GT. MAXIT) THEN
      WRITE(1,90)
      WRITE(33,90)
      STOP
      END IF
      FOLD = B + A*ALOG(TOLD) - TOLD
      FPOLD = (A - TOLD)/TOLD
      TNEW = TOLD - FOLD/FPOLD
      IF(ABS(TNEW-TOLD) .GT. 0.0000100) THEN
```

```
      TOLD = TNEW
      GO TO 12
      ELSE
      TRANS = 86400.0*TNEW
      END IF
      HYCON = TRANS/THICK
      WRITE(33,92) HYCON
      WRITE(1,92) HYCON
      TRANS = TRANS/86400.0
C
C     Test condition for alpha using 0.02 as threshold
C
      ALFA = 0.890*RS*SQRT(STO/TRANS)*((OMEGA**2 + GAMMA**2)**0.250)
      IF(ALFA .GE. 0.020) THEN
      WRITE(1,96) ALFA
      WRITE(33,96) ALFA
      END IF
C
C     The algorithm for the overdamped response begins here
C
      ELSE IF(ID .EQ. 2) THEN
      WRITE(1,*) 'APPLYING BOUWER AND RICE METHOD ... '
      WRITE(33,*) 'APPLYING BOUWER AND RICE METHOD ... '
C
C     Water-level equilibrium is shifted to zero
C     time unit is converted from minutes to seconds
C
      DO 20  J=1,INDEX
      READ(31,*) XVAL(J),YVAL(J)
      YVAL(J) = YVAL(J) - REFH
      IF(J .EQ. 1) THEN
      YCOMP = YVAL(1)/EXP(1.0)
      END IF
      IF(ABS(YVAL(J)) .GE. ABS(YCOMP)) THEN
      ISAVE = J
      END IF
      XVAL(J) = 60.0*XVAL(J)
      YVAL(J) = ALOG(ABS(YVAL(J)))
20    CONTINUE
C
C     A linear regression is done over one natural logarithm cycle
C
      CALL RLINE(ISAVE, XVAL, YVAL, B0, B1, STAT)
      WRITE(1,106) ISAVE
      WRITE(33,106) ISAVE
C
      XA = RLBR/RS
C
C     Read cubic spline points from fig. 3 in Bouwer & Rice paper
C     and perform the cubic spline interpolation
C
      DO 22  I=1,NDAT
      READ(32,*) XD(I),FDA(I),FDB(I),FDC(I)
22    CONTINUE
      CALL CSAKM(NDAT, XD, FDA, BREAK, CSCOEF)
      YA = CSVAL(XA, NINTV, BREAK, CSCOEF)
C
      CALL CSAKM(NDAT, XD, FDB, BREAK, CSCOEF)
      YB = CSVAL(XA, NINTV, BREAK, CSCOEF)
C
      CALL CSAKM(NDAT, XD, FDC, BREAK, CSCOEF)
      YC = CSVAL(XA, NINTV, BREAK, CSCOEF)
C
C     Obtain the value of hydraulic conductivity from equation 3
C
```

```
      IF(H .EQ. DBR) THEN
      RLNQ = 1.0/(1.10/ALOG(H/RS) + YC*RS/RLBR)
      ELSE
        IF(ALOG((DBR - H)/RS) .GE. 6.0) THEN
        TERM = 6.0
        ELSE
        TERM = ALOG((DBR - H)/RS)
        END IF
      RLNQ = 1.0/(1.10/ALOG(H/RS) + RS*(YA + YB*TERM)/RLBR)
      END IF
C
      HYDCON = -43200.0*RC*RC*RLNQ*B1/RLBR
      WRITE(33,94) ABS(HYDCON)
      WRITE(1,94)  ABS(HYDCON)
C
      END IF
C
90    FROMAT(' NUMBER OF ITERATIONS IN NEWTON'S METHOD EXCEEDED')
92    FORMAT(' K  FROM VAN DER KAMP METHOD IS ',F8.0,' FT/DAY')
94    FORMAT(' K  FROM BOUWER AND RICE METHOD IS ',F8.1,' FT/DAY')
96    FORMAT(' WARNING ... VALUE OF ALFA IS ',F8.3)
98    FORMAT(' S MAX = ',F7.4,'   S FREQ = ',F7.4,'   S TIME = ',F7.4)
100   FORMAT(' F MAX = ',F7.4,'   F FREQ = ',F7.4,'   F TIME = ',F7.4)
102   FORMAT(' DAMPING VALUES ARE: ',4(F7.4,2X))
104   FORMAT(' ANGULAR FREQUENCY VALUES ARE: ',6(F7.4,2X))
106   FORMAT(' NUMBER OF POINTS USED IN LINEAR REGRESSION: ',I2)
      STOP
      END
C
      SUBROUTINE FCN(X,F,N)
      REAL X(3),F(3),HVAL(3),TVAL(3)
      COMMON /VAL/ HVAL,TVAL
C
C     This subroutine evaluates equation (4) in paper for each one
C     of the three components
C
      DO 2 K=1,N
      F(K) = HVAL(K) - X(1)*COS(X(2)*(TVAL(K)-X(3)))
2     CONTINUE
      RETURN
      END
C
      SUBROUTINE LSJAC(N,X,FJAC)
      REAL X(3),FJAC(3,3),HVAL(3),TVAL(3)
      COMMON /VAL/ HVAL,TVAL
C
C     This subroutine evaluates the 3 by 3 jacobian matrix derived
C     from the nonlinear system of equations evaluated in FCN.
C     Subroutines FCN and LSJAC are externally called by the IMSL package.
C
      FJAC(1,1) = COS(X(2)*(TVAL(1)-X(3)))
      FJAC(1,2) = -X(1)*(TVAL(1)-X(3))*SIN(X(2)*(TVAL(1)-X(3)))
      FJAC(1,3) = X(1)*X(2)*SIN(X(2)*(TVAL(1)-X(3)))
      FJAC(2,1) = COS(X(2)*(TVAL(2)-X(3)))
      FJAC(2,2) = -X(1)*(TVAL(2)-X(3))*SIN(X(2)*(TVAL(2)-X(3)))
      FJAC(2,3) = X(1)*X(2)*SIN(X(2)*(TVAL(2)-X(3)))
      FJAC(3,1) = COS(X(2)*(TVAL(3)-X(3)))
      FJAC(3,2) = -X(1)*(TVAL(3)-X(3))*SIN(X(2)*(TVAL(3)-X(3)))
      FJAC(3,3) = X(1)*X(2)*SIN(X(2)*(TVAL(3)-X(3)))
      RETURN
      END
```

Appendix 2. SLUG.IN2 input data file

1.0	1.700	0.250	0.800
4.0	1.700	0.250	0.800
5.0	1.730	0.250	0.867
6.0	1.760	0.250	0.933
7.0	1.790	0.250	1.000
8.0	1.820	0.250	1.100
9.0	1.850	0.250	1.200
10.0	1.880	0.250	1.250
15.0	2.000	0.300	1.500
20.0	2.150	0.325	1.700
25.0	2.300	0.350	1.900
30.0	2.450	0.375	2.100
35.0	2.600	0.400	2.300
40.0	2.750	0.450	2.500
45.0	2.900	0.475	2.600
50.0	3.050	0.500	2.750
60.0	3.300	0.550	3.000
70.0	3.550	0.600	3.300
80.0	3.800	0.650	3.600
90.0	4.100	0.700	3.850
100.0	4.400	0.750	4.200
110.0	4.600	0.790	4.460
120.0	4.800	0.830	4.720
130.0	5.000	0.870	4.980
140.0	5.200	0.910	5.240
150.0	5.400	0.950	5.500
175.0	5.700	1.0375	6.125
200.0	6.000	1.125	6.750
225.0	6.300	1.2375	7.350
250.0	6.600	1.350	7.950
275.0	6.825	1.425	8.3250
300.0	7.050	1.500	8.700
325.0	7.225	1.600	9.025
350.0	7.400	1.700	9.350
375.0	7.550	1.7875	9.5750
400.0	7.700	1.875	9.800
425.0	7.775	1.9375	10.000
450.0	7.850	2.000	10.200
475.0	7.925	2.0625	10.350
500.0	8.000	2.125	10.500
550.0	8.200	2.250	10.8250
600.0	8.400	2.375	11.150
650.0	8.500	2.4375	11.325
700.0	8.600	2.500	11.500
750.0	8.700	2.575	11.625
800.0	8.800	2.650	11.750
850.0	8.875	2.700	11.875
900.0	8.950	2.750	12.000
950.0	9.000	2.800	12.100
1000.0	9.050	2.850	12.200
1100.0	9.140	2.905	12.320
1200.0	9.230	2.960	12.440
1300.0	9.320	3.015	12.560
1400.0	9.410	3.070	12.680
1500.0	9.500	3.125	12.800